/* $OpenBSD: setjmp.S,v 1.8 2005/08/07 11:30:38 espie Exp $ */
/*-
 * Copyright (c) 1990 The Regents of the University of California.
 * All rights reserved.
 *
 * This code is derived from software contributed to Berkeley by
 * William Jolitz.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions
 * are met:
 * 1. Redistributions of source code must retain the above copyright
 *    notice, this list of conditions and the following disclaimer.
 * 2. Redistributions in binary form must reproduce the above copyright
 *    notice, this list of conditions and the following disclaimer in the
 *    documentation and/or other materials provided with the distribution.
 * 3. Neither the name of the University nor the names of its contributors
 *    may be used to endorse or promote products derived from this software
 *    without specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE REGENTS AND CONTRIBUTORS ``AS IS'' AND
 * ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
 * ARE DISCLAIMED.  IN NO EVENT SHALL THE REGENTS OR CONTRIBUTORS BE LIABLE
 * FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL
 * DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS
 * OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION)
 * HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT
 * LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY
 * OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF
 * SUCH DAMAGE.
 */

#include <machine/asm.h>

/*
 * C library -- setjmp, longjmp
 *
 *	longjmp(a,v)
 * will generate a "return(v)" from the last call to
 *	setjmp(a)
 * by restoring registers from the stack.
 * The previous signal state is restored.
 */

#define NACL_BLOCK_SHIFT        4

    .text
    .align NACL_BLOCK_SHIFT
    .globl setjmp

setjmp:
  push  %ebp
  mov   %esp, %ebp
  push  %edi
  mov   0x8(%ebp), %edi
  mov   %eax, 0x00(%edi)
  mov   %ebx, 0x04(%edi)
  mov   %ecx, 0x08(%edi)
  mov   %edx, 0x0C(%edi)
  mov   %esi, 0x10(%edi)
  mov   -0x4(%ebp), %eax
  mov   %eax, 0x14(%edi)
  mov   0x00(%ebp), %eax
  nop
  nop
  mov   %eax, 0x18(%edi)
  mov   %esp, %eax
  add   $0xC, %eax
  mov   %eax, 0x1C(%edi)
  mov   0x04(%ebp), %eax
  mov   %eax, 0x20(%edi)
  pop   %edi
  mov   $0x0,%eax
  leave
  pop   %ecx
  and   $0xFFFFFFE0, %ecx
  jmp   *%ecx


    .text
    .align NACL_BLOCK_SHIFT
    .globl longjmp

longjmp:
  push  %ebp
  mov    %esp,%ebp
  mov    0x8(%ebp),%edi
  mov    0xc(%ebp),%eax
  test   %eax,%eax
  jne    0f
  inc    %eax
0:
  mov    %eax,(%edi)
  mov    0x18(%edi),%ebp
  mov    0x1c(%edi),%esp
  pushl  0x20(%edi)
  mov    (%edi),%eax
  mov    0x4(%edi),%ebx
  nop
  nop
  mov    0x8(%edi),%ecx
  mov    0xc(%edi),%edx
  mov    0x10(%edi),%esi
  mov    0x14(%edi),%edi
  pop    %ecx
  and    $0xffffffe0,%ecx
  jmp    *%ecx
